Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Chemical Engineering



INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁎. When you see ⁎ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^{T}SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^{T}THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^{T} ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^{T}R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

* Requires one additional memory module

STRAIGHT FIN EFFICIENCY

This program evaluates fin efficiency. Given the number of fins per unit of surface length and the temperature difference, the total heat transfer can also be found.

Equations:

$$\eta_f = \frac{\tanh\ (y)}{y}$$

$$y = (L + x/2)^{3/2} \sqrt{2h/kxL}$$

$$q = h[(1 - N_{ave} x) + \eta_f N_{ave} (2L + x)] \Delta T$$

$$\Delta T = |T_o - T_1|$$



where η_f is fin efficiency;

L is fin length;

x is fin thickness;

h is the convective coefficient;

k is the conductive coefficient;

N_{ave} is the average number of fins per unit length of surface area;

q is the total heat flux per unit area;

T_o is the temperature of the base of the fin;

T_1 is the fluid temperature.

Remarks:

Dimensional consistency must be maintained.

Example:

The oil pan of a race car is to be cooled by adding aluminum fins (k = 133 Btu/hr - °F - ft). The convective coefficient is about 50 Btu/hr - °F - ft^2. The fins are to be 0.1 inch thick, 0.5 inches long and average 15 per foot. If T_o is taken to be 300°F and T_1 is 110°F, what is the total heat transfer? What is the heat transfer without any fins?

Keystrokes:	Display:	
[USER]		(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 008		
[XEQ] [ALPHA] FIN [ALPHA]	H?	
50 [R/S]	K?	
133 [R/S]	X?	
0.1 [ENTER↑] 12 [÷] [R/S]	L?	
0.5 [ENTER↑] 12 [÷] [R/S]	NF=0.94	
[R/S]	NA?	
15 [R/S]	DELTA T?	
300 [ENTER↑] 110 [-] [R/S]	Q=20,537.02	
[A]	NA?	
0 [R/S]	DELTA T?	
300 [ENTER↑] 110 [-] [R/S]	Q=9,500.00	

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode		[USER]	
2	Begin execution and input data		[XEQ] FIN	H?
	convective coefficient	h	[R/S]	K?
	conductive coefficient	k	[R/S]	X?
	fin thickness	x	[R/S]	L?
	fin length	L	[R/S]	NF=
			[R/S]	NA?
		N_{ave}	[R/S]	DELTA T?
		ΔT	[R/S]	Q=
3	To make a change in N_{ave} and/or ΔT		[A]	NA?
		N_{ave}	[R/S]	DELTA T?
		ΔT	[R/S]	Q=

Program Listings

Initialization

```
01♦LBL "FIN"
02 FIX 2
03 DEG
04 "H?"
05 PROMPT
06 STO 03
07 "K?"
08 PROMPT
09 STO 02
10 "X?"
11 PROMPT
12 2
13 /
14 STO 01
15 "L?"
16 PROMPT
17 STO 04
```

η_f

```
18 RCL 01
19 +
20 SQRT
21 LASTX
22 *
23 RCL 03
24 RCL 02
25 /
26 RCL 01
27 /
28 RCL 04
29 /
30 SQRT
31 *
32 STO 07
33 E↑X
34 ATAN
35 ENTER↑
36 +
37 90
38 -
39 SIN
40 RCL 07
41 /
42 STO 05
43 "NF="
44 XEQ d
```

Q

```
45♦LBL A
46 "NA?"
47 PROMPT
48 STO 06
49 RCL 01
50 ENTER↑
51 +
52 *
53 1
54 X<=Y?
55 0
56 /
57 "DELTA T?"
58 PROMPT
59 RCL 04
60 RCL 01
61 +
62 ENTER↑
63 +
64 RCL 06
65 *
66 RCL 05
67 *
68 RCL 01
69 ENTER↑
70 +
71 RCL 06
72 *
73 CHS
74 1
75 +
76 +
77 *
78 RCL 03
79 *
80 "Q="
```

Display

```
81♦LBL d
82 ARCL X
83 AVIEW
84 STOP
85 RTN
86 .END.
```

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	x/2		
	K		
	h		
	L		
05	n_f	55	
	N_{ave}		
	temp. storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 025 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		NONE	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

STRAIGHT FIN EFFICIENCY

PROGRAM REGISTERS NEEDED: 18

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (5 - 13)

ROW 3 (14 - 24)

ROW 4 (25 - 37)

ROW 5 (37 - 44)

ROW 6 (45 - 53)

ROW 7 (54 - 58)

ROW 8 (59 - 71)

ROW 9 (72 - 81)

ROW 10 (82 - 86)

CONSERVATION OF ENERGY

This program converts kinetic energy, potential energy and pressure-volume work to energy.

Equation:

$$\frac{V_1^2}{2} + gz_1 + \frac{P_1}{\rho} + \frac{E_1}{\dot{m}} = \frac{V_2^2}{2} + gz_2 + \frac{P_2}{\rho} + \frac{E_2}{\dot{m}}$$

where V is the fluid velocity;

z is the height above a reference datum;

P is the pressure;

E is an energy term which could represent inputs of work or friction loses (negative value);

g is the acceleration of gravity;

ρ is the fluid density;

$\dot{m}$ is the mass flow rate (assumed to be unity);

subscripts $_1$ and $_2$ refer to upstream and downstream values respectively.

Example:

A water tower is 100 feet high. What is the zero flow rate pressure at the base? The density of water is 62.4 lb/ft^3.

Keystrokes:	Display:	
[USER]		(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009		
[XEQ] [ALPHA] ENG [ALPHA]	DENSITY?	
62.4 [R/S]	V?	
0 [R/S]	Z?	
100 [R/S]	P?	
[R/S]	E?	
0 [R/S]		
[C]	P=43.33	

If water is flowing out of the tower at a velocity of 10 ft/sec, what is the static pressure?

Keystrokes:	Display:
[XEQ] [ALPHA] ENG [ALPHA]	DENSITY?
62.4 [R/S]	V?
10 [CHS] [R/S]	Z?
100 [R/S]	P?
[R/S]	E?
0 [R/S]	
[C]	P=42.66

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode		[USER]	
2	Begin execution:			
	for English units		[XEQ] ENG	DENSITY?
	for SI units		[XEQ] SI	DENSITY?
3	Input data†	ρ	[R/S]	V?
		V	[R/S]	Z?
		z	[R/S]	P?
		P	[R/S]	E?
		E	[R/S]	
4	To find:††			
	velocity		[A]	V=
	height		[B]	Z=
	pressure		[C]	P=
	energy		[D]	E=
	†Downstream values should be input as negatives. However, when an output is called for, the calculator displays the relative value with no regard to upstream or downstream location.			
	††"DATA ERROR" will result when the total energy sum stored in register 8 is negative and an attempt is made to calculate velocity.			

Program Listings

`01♦LBL "SI"`	SI constants
`02 CF 00`	
`03 "DENSITY?"`	
`04 PROMPT`	
`05 STO 04`	
`06 CLX`	
`07 STO 08`	
`08 9.80665`	
`09 STO 06`	
`10 GTO 00`	
`11♦LBL "ENG"`	English constants
`12 SF 00`	
`13 "DENSITY?"`	
`14 PROMPT`	
`15 STO 04`	
`16 CLX`	
`17 STO 08`	
`18 778.16`	
`19 STO 05`	
`20 32.17`	
`21 STO 06`	
`22♦LBL 00`	Initialization
`23 CLX`	
`24 "V?"`	
`25 PROMPT`	
`26 ENTER↑`	
`27 ABS`	
`28 *`	
`29 2`	
`30 /`	
`31 ST+ 08`	
`32 CLX`	
`33 "Z?"`	
`34 PROMPT`	
`35 ENTER↑`	
`36 RCL 06`	
`37 *`	
`38 ST+ 08`	
`39 CLX`	
`40 "P?"`	
`41 PROMPT`	
`42 ENTER↑`	
`43 FC? 00`	
`44 GTO 00`	
`45 144`	
`46 STO 07`	
`47 *`	
`48♦LBL 00`	
`49 RCL 04`	
`50 /`	
`51 FC? 00`	
`52 GTO 00`	
`53 RCL 06`	
`54 *`	
`55♦LBL 00`	
`56 ST+ 08`	
`57 CLX`	
`58 "E?"`	
`59 PROMPT`	
`60 ENTER↑`	
`61 FC? 00`	
`62 GTO 00`	
`63 RCL 05`	
`64 *`	
`65 RCL 06`	
`66 *`	
`67♦LBL 00`	
`68 ST+ 08`	
`69 STOP`	
`70♦LBL A`	Velocity
`71 RCL 08`	
`72 2`	
`73 *`	
`74 SQRT`	
`75 "V="`	
`76 GTO d`	
`77♦LBL B`	Height
`78 RCL 08`	
`79 RCL 06`	
`80 /`	
`81 "Z="`	
`82 GTO d`	
`83♦LBL C`	Pressure
`84 RCL 08`	
`85 FC? 00`	
`86 GTO 00`	
`87 RCL 07`	
`88 /`	
`89♦LBL 00`	
`90 RCL 04`	
`91 *`	
`92 "P="`	
`93 FC? 00`	
`94 GTO d`	
`95 RCL 06`	
`96 /`	
`97 GTO d`	
`98♦LBL D`	Energy
`99 "E="`	

Program Listings

```
100 RCL 08
101 FC? 00
102 GTO d
103 RCL 05
104 /
105 RCL 06
106 /
107♦LBL d
108 ARCL X
109 AVIEW
110 STOP
111 .END.
```

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	ρ		
05	778.16	55	
	g		
	114		
	ΣE		
	temp storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 038 USER MODE
ENG ___ FIX 2 SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		ENGLISH	SI

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CONSERVATION OF ENERGY

PROGRAM REGISTERS NEEDED: 30

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 3)

ROW 2 (3 - 8)

ROW 3 (8 - 12)

ROW 4 (12 - 16)

ROW 5 (17 - 20)

ROW 6 (21 - 31)

ROW 7 (31 - 40)

ROW 8 (40 - 47)

ROW 9 (48 - 57)

ROW 10 (58 - 66)

ROW 11 (67 - 75)

ROW 12 (76 - 82)

ROW 13 (82 - 91)

ROW 14 (92 - 97)

ROW 15 (98 - 104)

ROW 16 (105 - 111)

HYDROCARBON COMBUSTION

Given the atomic composition of a hydrocarbon fuel and the desired amount of excess air, the program calculates the air-fuel ratio on a mass and mole basis. The number of moles of products is also calculated along with the volume percents of sulfur dioxide, carbon dioxide, water vapor, oxygen and nitrogen. Complete combustion is assumed.

Equations:

Total combustion air

$$AIR = 1 + \frac{\%\ \text{Excess Air}}{100}$$

Consumed oxygen

$$O_2 = C + S + \frac{H}{4} - \frac{O}{2}$$

Air-fuel ratios

$$AF(\text{mole}) = O_2(4.762)\ AIR$$

$$AF(\text{mass}) = \frac{1.8094\ AF\ (\text{mole})}{0.7507C + 0.063H + 2.004S + 0.875N + O}$$

Total moles of products

$$M = O_2[4.762\ AIR] + \frac{H}{4} + \frac{O}{2} + \frac{N}{2}$$

Volume percents of products

$$\text{Volume}\ \%CO_2 = \frac{100C}{M}$$

$$\text{Volume}\ \%SO_2 = \frac{100S}{M}$$

$$\text{Volume}\ \%H_2O = \frac{100H}{2M}$$

$$\text{Volume}\ \%O_2 = \frac{100(AIR - 1)\ O_2}{M}$$

$$\text{Volume}\ \%N_2 = \frac{(100)\left[(3.762)\ AIR\ O_2 + \frac{N}{2}\right]}{M}$$

where C, S, N, H and O refer to the number of carbon, sulfur, nitrogen, hydrogen and oxygen atoms respectively per hypothetical fuel molecule;

AF stands for air-fuel ratio;

M stands for total moles of product.

Remarks:

% Excess air ≥ 0.

Complete combustion is assumed.

The volume percent values assume that no water vapor has been condensed out.

Example:

Octane, C_8H_{18}, is burned in 40% excess air. What is the air-fuel mass ratio and what are the volume percents of the products?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] HCC [ALPHA]	C?
8 [R/S]	H?
18 [R/S]	O?
0 [R/S]	S?
0 [R/S]	N?
0 [R/S]	%EXCESS AIR?
40 [R/S]	AF,MASS=21.12
[R/S]	AF,MOLES=83.34
[R/S]	PROD MOLES=87.84
[R/S]	%SO2=0.00
[R/S]	%CO2=9.11
[R/S]	%H2O=10.25
[R/S]	%O2=5.69
[R/S]	%N2=74.95

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution and input data		[XEQ] HCC	C?
	i) atomic composition (enter zero if	C	[R/S]	H?
	element is not present)	H	[R/S]	O?
		O	[R/S]	S?
		S	[R/S]	N?
		N	[R/S]	%EXCESS AIR?
	ii) percent excess air	% EXCESS AIR	[R/S]	
3	Solutions			AF,MASS=
			[R/S]	AF,MOLES=
			[R/S]	PROD MOLES=
			[R/S]	%SO2=
			[R/S]	%CO2=
			[R/S]	%H2O=
			[R/S]	%O2=
			[R/S]	%N2=

Program Listings

Initialization

```
01♦LBL "HCC
"
02 FIX 2
03 "C?"
04 PROMPT
05 STO 01
06 "H?"
07 PROMPT
08 STO 02
09 "O?"
10 PROMPT
11 STO 03
12 "S?"
13 PROMPT
14 STO 04
15 "N?"
16 PROMPT
17 STO 09
18 "%EXCESS
AIR?"
19 PROMPT
```

Air

```
20 100
21 /
22 1
23 +
24 STO 05
```

O_2

```
25 RCL 01
26 RCL 04
27 +
28 RCL 02
29 4
30 /
31 +
32 RCL 03
33 2
34 /
35 -
36 STO 06
```

AF (mole)

```
37 *
38 4.762
39 *
40 STO 08
```

M

```
41 RCL 02
42 4
43 /
44 +
45 RCL 03
46 2
47 /
48 +
49 RCL 09
50 2
51 /
52 +
53 STO 07
```

AF (mass)

```
54 RCL 08
55 1.8094
56 *
57 RCL 01
58 .7507
59 *
60 RCL 02
61 .063
62 *
63 +
64 RCL 03
65 +
66 2.004
67 RCL 04
68 *
69 +
70 .875
71 RCL 09
72 *
73 +
74 /
```

Outputs

```
75 "AF,MASS
="
76 XEQ d
77 RCL 08
78 "AF,MOLE
S="
79 XEQ d
80 RCL 07
81 "PROD MO
LES="
82 XEQ d
83 RCL 04
84 XEQ E
85 "%SO2="
86 XEQ d
87 RCL 01
88 XEQ E
89 "%CO2="
90 XEQ d
91 RCL 02
92 2
93 /
94 XEQ E
95 "%H2O="
96 XEQ d
97 RCL 05
```

Program Listings

```
 98 1
 99 -
100 RCL 06
101 *
102 XEQ E
103 "%O2="
104 XEQ d
105 RCL 08
106 RCL 05
107 RCL 06
108 *
109 -
110 RCL 09
111 2
112 /
113 +
114 XEQ E
115 "%N2="
116♦LBL d
117 ARCL X
118 AVIEW
119 STOP
120 RTN
121♦LBL E
122 RCL 07
123 /
124 100
125 *
126 RTN
127 .END.
```

Display (steps 116–120)

Subroutine (steps 121–127)

$\frac{100}{M}$ x

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	C		
	H		
	O		
	S		
05	AIR	55	
	O_2		
	Prod		
	AF(moles)		
	N		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 047 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		NONE	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HYDROCARBON COMBUSTION

HEWLETT PACKARD
SOLUTIONS BOOK
CHEMICAL ENGINEERING

PROGRAM REGISTERS NEEDED: 38

ROW 1 (1 : 4)

ROW 2 (5 : 12)

ROW 3 (12 : 18)

ROW 4 (18 : 21)

ROW 5 (22 : 34)

ROW 6 (35 : 43)

ROW 7 (44 : 55)

ROW 8 (55 : 60)

ROW 9 (61 : 66)

ROW 10 (67 : 75)

ROW 11 (75 : 78)

ROW 12 (78 : 81)

ROW 13 (81 : 82)

ROW 14 (82 : 86)

ROW 15 (86 : 90)

ROW 16 (90 : 95)

ROW 17 (96 : 103)

ROW 18 (103 : 111)

ROW 19 (112 : 117)

ROW 20 (117 : 126)

ROW 21 (127 : 127)

HEAT TRANSFER THROUGH COMPOSITE CYLINDERS AND WALLS

This program can be used to calculate the overall heat transfer coefficient for composite tubes and walls from individual section conductances and surface coefficients.



Figure 1.—Composite tube



Figure 2. —Composite wall

Equations:

The overall heat transfer coefficient U is defined by:

$$q/L = U\ \Delta T$$

or

$$q/A = U\ \Delta T$$

where ΔT is the total temperature difference $(T_2 - T_1)$, q/L is the heat transfer per unit length of pipe, and q/A is the heat transfer per unit area of wall.

For cylinders

$$U = \frac{2\pi}{\dfrac{2}{h_1 D_1} + \dfrac{\ln(D_2/D_1)}{k_1} + \dfrac{\ln(D_3/D_2)}{k_2} + \ldots + \dfrac{\ln(D_n/D_{n-1})}{k_{n-1}} + \dfrac{2}{h_n D_n}}$$

For walls

$$U = \frac{1}{\frac{1}{h_1} + \frac{x_1}{k_1} + \frac{x_2}{k_2} + \ldots + \frac{x_n}{k_n} + \frac{1}{h_n}}$$

where

h is the convective surface coefficient;

D_n is the outside diameter of the annulus;

k is the conductive coefficient;

x is the thickness of a wall section.

Remarks:

These equations are for steady state heat transfer through materials with constant properties in all directions.

For composite cylinders, inputs must start with the inside convective coefficient and work out.

Zero is an invalid input for D, k, and h.

Dimensional consistency must be maintained.

Example:

A steel pipe with an inside diameter of 4 inches and a thickness of 0.5 inches has a conductivity of 25 Btu/ft-hr-°F. Two inches of asbestos (k = 0.1 Btu/hr-ft-°F) enclose the pipe bringing the total diameter to 9 inches. If the inside convective coefficient is 1000 Btu/hr-ft^2-°F and the outside coefficient is 5 Btu/hr-ft^2-°F, what is the overall heat transfer coefficient? What is the heat loss for 100 feet of pipe if ΔT is 115°F?

Keystrokes:	Display:	
[XEQ] [ALPHA] SIZE [ALPHA] 009		
[XEQ] [ALPHA] CYL [ALPHA]	NO. OF SECTS?	
2 [R/S]	D?	
4 [ENTER↑] 12 [÷] [R/S]	H?	
1000 [R/S]	D?	
5 [ENTER↑] 12 [÷] [R/S]	K?	
25 [R/S]	D?	
9 [ENTER↑] 12 [÷] [R/S]	K?	
0.1 [R/S]	H?	
5 [R/S]	U=0.98	Btu/hr-ft-°F
115 [X]	112.44	Btu/hr-ft
100 [X]	11,244.20	Btu/hr

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	For walls go to step 3, and for cylinders			
	go to step 9.			
3	Initialize for walls		[XEQ] WALLS	NO. OF SECTS?
4	Key in number of sections	N	[R/S]	H?
5	Key in the first section's			
	convective coefficient	h_1	[R/S]	X?
6	Key in thickness of the current section	x_i	[R/S]	k?
7	Key in the conductive coefficient for the			
	section of step 6	k_i	[R/S]	X? or H?
	(Repeat steps 6 and 7 for each section.			
	The prompt after the last section will be			
	"H?")			
8	Key in the last section's			
	convective coefficient	h_n	[R/S]	U=
9	Initialize for cylinders		[XEQ] CYL	NO. OF SECTS?
10	Key in number of sections	N	[R/S]	D?
11	Key in the inside section's inner			
	diameter	D_1	[R/S]	H?
12	Key in the inside convective coefficient	h_1	[R/S]	D?
13	Key in the outside diameter of the current			
	section	D_i	[R/S]	K?
14	Key in the conductive coefficient for the			
	section of step 13	k_i	[R/S]	D? or H?
	(Repeat steps 13 and 14 for each section)			
15	Key in the outside convective coefficient	h_n	[R/S]	U=

Program Listings

Listing	Comments
`01♦LBL "CYL"` `02 SF 00` `03 GTO 00` `04♦LBL "WALLS"` `05 CF 00` `06♦LBL 00` `07 FIX 2` `08 1` `09 FS? 00` `10 PI` `11 STO 06` `12 CLX` `13 STO 08` `14 FC? 00` `15 GTO 01`	Initialization
`16 "NO. OF SECTS?"` `17 PROMPT` `18 STO 00` `19 "D?"` `20 PROMPT` `21♦LBL A` `22 STO 07` `23 "H?"` `24 PROMPT` `25 *` `26 1/X` `27 ST+ 08` `28 FC?C 00` `29 GTO "U"` `30♦LBL B` `31 "D?"` `32 PROMPT` `33 STO 01` `34 "K?"` `35 PROMPT` `36 1/X` `37 X<>Y` `38 RCL 07` `39 X<>Y` `40 STO 07` `41 /` `42 LN` `43 *` `44 2` `45 /` `46 ST- 08` `47 DSE 00` `48 GTO B` `49 RCL 01` `50 GTO A`	Cylinders
`51♦LBL 01` `52 SF 00` `53 "NO. OF SECTS?"` `54 PROMPT` `55 STO 00` `56♦LBL C` `57 "H?"` `58 PROMPT` `59 1/X` `60 ST+ 08` `61 FC?C 00` `62 GTO "U"` `63♦LBL D` `64 "X?"` `65 PROMPT` `66 "K?"` `67 PROMPT` `68 /` `69 ST+ 08` `70 DSE 00` `71 GTO D` `72 GTO C`	Walls
`73♦LBL "U"` `74 RCL 08` `75 1/X` `76 RCL 06` `77 *` `78 STO 04` `79 "U="` `80 ARCL X` `81 AVIEW` `82 STOP` `83 .END.`	Calculate U and display

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	No. of surfaces	50	
	outside diameter		
	U		
05		55	
	1 or π		
	temp. storage		
	ΣR		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 034 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Cyl	Walls

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEAT TRANSFER THROUGH
COMPOSITE CYLINDERS AND WALLS
PROGRAM REGISTERS NEEDED: 26

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 – 4)

ROW 2 (4 – 8)

ROW 3 (9 – 16)

ROW 4 (16 – 18)

ROW 5 (19 – 26)

ROW 6 (27 – 32)

ROW 7 (33 – 43)

ROW 8 (44 – 50)

ROW 9 (51 – 53)

ROW 10 (53 – 59)

ROW 11 (60 – 65)

ROW 12 (66 – 72)

ROW 13 (72 – 79)

ROW 14 (79 – 83)

VON KÁRMÁN ANALOGY FOR HEAT AND MASS TRANSFER

The von Kármán analogy forms a link between momentum, heat, and mass transfer for conduit flow. If any of the transport coefficients f, h, or k_c are known, the others can be found using this program and the appropriate fluid transport numbers.

Equations:

Heat transfer

$$St = \frac{f/2}{1 + 5\sqrt{f/2}\,(Pr - 1 + \ln[1 + 5/6\,(Pr - 1)])}$$

Mass transfer

$$\frac{k_c}{v} = \frac{f/2}{1 + 5\sqrt{f/2}\,(Sc - 1 + \ln[1 + 5/6\,(Sc - 1)])}$$

In both cases, the equation for f is solved using Newton's method, with the Colburn analogy as the initial guess, f_o.

$$\frac{f_o}{2} = St\,(Pr)^{2/3} = \frac{k_c}{v}\,(Sc)^{2/3}$$

where

St is the Stanton number.

f is the Fanning friction factor.

Pr is the Prandtl number.

Sc is the Schmidt number.

k_c is the convective mass transfer coefficient.

v is the average fluid velocity.

Reference:

Welty, Wicks, Wilson, *Fundamentals of Momentum Heat and Mass Transfer*; John Wiley and Sons, Inc., 1969.

Remarks:

No form drag may be present. Fanning friction factors should be less than 0.02 and greater than 0.0001.

Example:

The Schmidt number for a mild acid flowing through a metal pipe has been found to be 3.7. The Fanning friction factor is 0.011. If the fluid velocity is 15 feet per second, what is the convective mass transfer coefficient?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] ANALOG [ALPHA]	SC? OR PR?
3.7 [R/S]	ST?
[R/S]	KC/V?
[R/S]	F?
0.011 [R/S]	
[XEQ] [ALPHA] KC/V [ALPHA]	KC/V=0.0023
15 [X]	0.0338

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution and input data		[XEQ] ANALOG	SC? OR PR?
		Sc or Pr	[R/S]	ST?
	Only one of St, k_c/v and f are needed.	St	[R/S]	KC/V?
	Ignore the others by pressing [R/S]	k_c/v	[R/S]	F?
		f	[R/S]	
3	To find:			
	St		[XEQ] ST	ST=
	k_c/v		[XEQ] KC/V	KC/V=
	f		[XEQ] F	F=

Program Listings

Initialization

```
 01+LBL "ANA
LOG"
 02 FIX 4
 03 "SC? OR
PR?"
 04 PROMPT
 05 STO 03
 06 "ST?"
 07 PROMPT
 08 "KC/V?"
 09 PROMPT
 10 STO 02
 11 "F?"
 12 PROMPT
 13 STO 05
 14 STOP
```

St

```
 15+LBL "ST"
 16 "ST="
 17 GTO B
```

k_c/v

```
 18+LBL "KC/
V"
 19 "KC/V="
```

St & k_c/v

```
 20+LBL B
 21 0
 22 RCL 05
 23 2
 24 /
 25 STO 06
 26 XEQ D
 27 /
 28 STO 02
 29 GTO d
```

f

```
 30+LBL F
 31 "F="
 32 RCL 05
 33 RCL 02
 34 RCL 03
 35 2
 36 ENTER↑
 37 3
 38 /
 39 Y↑X
 40 *
 41 STO 06
 42+LBL 01
 43 XEQ D
 44 RCL 02
 45 *
 46 STO 04
 47 RCL 06
 48 X<>Y
 49 -
 50 RCL 04
 51 RCL 02
 52 -
 53 RCL 06
 54 /
 55 2
 56 /
 57 1
 58 -
 59 /
 60 +
 61 STO 06
 62 -
 63 ABS
 64 1 E-8
 65 X<=Y?
 66 GTO 01
 67 RCL 06
 68 2
 69 *
 70 STO 05
 71 GTO d
 72+LBL D
 73 RCL 06
 74 RCL 03
 75 1
 76 -
 77 ENTER↑
 78 ENTER↑
 79 5
 80 *
 81 6
 82 /
 83 1
 84 +
 85 LN
 86 +
 87 STO 08
 88 X<>Y
 89 SQRT
 90 *
 91 5
 92 *
 93 1
 94 +
 95 RTN
```

Display

```
 96+LBL d
 97 ARCL X
 98 AVIEW
 99 STOP
100 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	St		
	Pr(Sc)		
	temp storage		
05	f	55	
	f/2		
	temp storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 33 USER MODE
ENG ____ FIX 4 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		NONE	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

VON KARMAN ANALOGY FOR HEAT AND MASS TRANSFER
PROGRAM REGISTERS NEEDED: 25

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 3)

ROW 2 (3 - 6)

ROW 3 (6 - 11)

ROW 4 (11 - 16)

ROW 5 (16 - 18)

ROW 6 (19 - 25)

ROW 7 (26 - 31)

ROW 8 (32 - 43)

ROW 9 (43 - 55)

ROW 10 (56 - 65)

ROW 11 (66 - 74)

ROW 12 (75 - 87)

RCW 13 (88 - 98)

ROW 14 (99 - 100)

EQUATIONS OF STATE

This program provides both ideal gas and Redlich-Kwong equations of state. Given four of the five state variables, the fifth is calculated. For the Redlich-Kwong solution, the critical pressure and temperature of the gas must be known. They are not needed for ideal gas solutions.

Values of the Universal Gas Constants

Value of R	Units of R	Units of P	Units of V	Units of T
8.314	N-m/g mole-K	N/m^2	m^3/g mole	K
83.14	cm^3-bar/g mole-K	bar	cm^3/g mole	K
82.05	cm^3-atm/g mole-K	atm	cm^3/g mole	K
0.7302	atm-ft^3/lb mole-°R	atm	ft^3/lb mole	°R
10.73	psi-ft^3/lb mole-°R	psi	ft^3/lb mole	°R
1545	psf-ft^3/lb mole-°R	psf	ft^3/lb mole	°R

Critical Temperatures and Pressures

Substance	T_c, K	T_c, °R	P_c, ATM
Ammonia	405.6	730.1	112.5
Argon	151	272	48.0
Carbon dioxide	304.2	547.6	72.9
Carbon monoxide	133	239	34.5
Chlorine	417	751	76.1
Helium	5.3	9.5	2.26
Hydrogen	33.3	59.9	12.8
Nitrogen	126.2	227.2	33.5
Oxygen	154.8	278.6	50.1
Water	647.3	1165.1	218.2
Dichlorodifluoromethane	384.7	692.5	39.6
Dichlorofluoromethane	451.7	813.1	51.0
Ethane	305.5	549.9	48.2
Ethanol	516.3	929.3	63
Methanol	513.2	923.8	78.5
n-Butane	425.2	765.4	37.5
n-Haxane	507.9	914.2	29.9
n-Pentane	469.5	845.1	33.3
n-Octane	568.6	1023.5	24.6
Trichlorofluoromethane	471.2	848.1	43.2

Equations:

Ideal gas: $PV = nRT$

Redlich-Kwong:

$$P = \frac{nRT}{(V - b)} - \frac{a}{T^{1/2} V (V + b)}$$

$$a = 4.934 \ b \ nRT_c^{1.5}$$

$$b = 0.0867 \frac{nRT_c}{P_c}$$

where:

P is the absolute pressure;

V is the volume;

n is the number of moles present;

R is the universal gas constant;

T is the absolute temperature;

T_c is the critical temperature;

P_c is the critical pressure.

Remarks:

P, V, n and T must have units compatible with R.

At low temperatures or high pressures, the ideal gas law does not represent the behavior of real gases.

No equation of state is valid for all substances or over an infinite range of conditions. The Redlich-Kwong equation gives moderate to good accuracy for a variety of substances over a wide range of conditions. Results should be used with caution and tempered by experience.

Solutions for V, n, R and T, using the Redlich-Kwong equation, require an iterative technique. Newton's method is employed using the ideal gas law to generate the initial guess. Iteration time is generally a function of the amount of deviation from ideal gas behavior. For extreme cases, the routine may fail to converge entirely, resulting in "DATA ERROR".

Example 1:

0.63 g moles of air are enclosed in a 25,000 cm^3 space at 1200 K. What is the pressure in bars? Assume ideal gas behavior.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[XEQ] [ALPHA] ID [ALPHA]	P?
0 [R/S]	V?
25000 [R/S]	N?
0.63 [R/S]	R?
83.14 [R/S]	T?
1200 [R/S]	P=2.51

Example 2:

The specific volume of a gas in a container is 800 cm^3/g mole. The temperature will reach 400 K. What will the pressure be, according to the Redlich-Kwong relation?

$$P_c = 48.2 \text{ atm}$$

$$T_c = 305.5 \text{ K}$$

$$R = 82.05 \text{ cm}^3 - \text{atm/g mole-K}$$

Keystrokes:	Display:
[XEQ] [ALPHA] RK [ALPHA]	TC?
305.5 [R/S]	PC?
48.2 [R/S]	P?
0 [R/S]	V?
800 [R/S]	N?
1 [R/S]	R?
82.05 [R/S]	T?
400 [R/S]	P=36.27

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Select equation:			
	Ideal gas		[XEQ] ID	P?
	Redlich-Kwong		[XEQ] RK	TC?
3	For Redlich-Kwong only	T_c	[R/S]	PC?
		P_c	[R/S]	P?
4	Input appropriate values	P	[R/S]	V?
	(Zero for the unknown)	V	[R/S]	N?
		n	[R/S]	R?
		R	[R/S]	T?
		T	[R/S]	(unknown) =

Program Listings

Initialization

```
 01♦LBL "ID"
 02 0
 03 SF 00
 04 GTO 00
 05♦LBL "RK"
 06 1
 07 CF 00
 08 "TC?"
 09 PROMPT
 10 STO 13
 11 "PC?"
 12 PROMPT
 13 STO 14
 14♦LBL 00
 15 SF 02
 16 CF 01
 17 FIX 2
 18 "P?"
 19 PROMPT
 20 5
 21 XEQ 00
 22 "V?"
 23 PROMPT
 24 6
 25 XEQ 00
 26 "N?"
 27 PROMPT
 28 7
 29 XEQ 00
 30 "R?"
 31 PROMPT
 32 8
 33 XEQ 00
 34 "T?"
 35 PROMPT
 36 CF 02
 37 9
 38♦LBL 00
 39 CF 01
 40 STO 01
 41 RDN
 42 STO IND
01
 43 X≠0?
 44 GTO 01
 45 R↑
 46 STO 10
 47 1
 48 STO IND
01
 49♦LBL 01
 50 FS? 02
 51 RTN
 52 GTO IND
10
```

Calculate unknown

```
 53♦LBL 05
 54 "P="
 55 GTO 00
 56♦LBL 06
 57 "V="
 58♦LBL 00
 59 RCL 07
 60 RCL 08
 61 *
 62 RCL 09
 63 *
 64 RCL 05
 65 RCL 06
 66 *
 67 /
 68 STO IND
10
 69 GTO 00
 70♦LBL 07
 71 SF 01
 72 "N="
 73 GTO 01
 74♦LBL 08
 75 SF 01
 76 "R="
 77 GTO 01
 78♦LBL 09
 79 "T="
 80 SF 01
 81♦LBL 01
 82 RCL 05
 83 RCL 06
 84 *
 85 RCL 07
 86 /
 87 RCL 08
 88 /
 89 RCL 09
 90 /
 91 STO IND
10
 92♦LBL 00
 93 FS? 00
 94 GTO 10
 95 XEQ 01
 96 GTO 00
 97♦LBL 02
```

Program Listings

If ideal, display

```
 98 FS? 01
 99 XEQ 01
```

```
100♦LBL 00
101 RCL 00
102 RCL 09
103 *
```

Calculate using Redlich-Kwong equations

```
104 RCL 06
105 RCL 12
106 -
107 STO 04
108 /
109 RCL 11
110 RCL 09
111 SQRT
112 /
113 STO 02
114 RCL 06
115 /
116 LASTX
117 RCL 12
118 +
119 STO 03
120 /
121 -
122 RCL 05
123 -
124 XEQ IND
10
125 /
126 ST- IND
10
127 RCL IND
10
128 /
129 ABS
130 1 E-4
131 X<=Y?
132 GTO 02
133 RCL IND
10
134 GTO 10
```

$\frac{\partial P}{\partial V}$

```
135♦LBL 06
136 RCL 06
137 ENTER↑
138 +
139 RCL 12
140 +
141 RCL 02
142 *
143 RCL 03
144 RCL 06
145 *
146 X↑2
147 /
148 RCL 00
149 RCL 09
150 *
151 RCL 04
152 X↑2
153 /
154 -
155 RTN
```

$\frac{\partial P}{\partial T}$

```
156♦LBL 09
157 RCL 00
158 RCL 04
159 /
160 RCL 02
161 2
162 /
163 RCL 09
164 /
165 RCL 06
166 /
167 RCL 03
168 /
169 +
170 RTN
```

$\frac{\partial P}{\partial n}$ or $\frac{\partial P}{\partial R}$

```
171♦LBL 07
172♦LBL 08
173 RCL 09
174 RCL 06
175 *
176 RCL 04
177 X↑2
178 /
179 RCL 06
180 ENTER↑
181 +
182 RCL 12
183 +
184 RCL 00
185 /
186 RCL 06
187 /
188 RCL 03
189 X↑2
190 /
191 RCL 02
192 *
193 -
194 RCL 00
195 *
196 RCL IND
```

Program Listings

```
 10
197 /
198 RTN
199♦LBL 05
200 LASTX
201 +
202 STO 05
203 GTO 10
```

Calculate a, b

```
204♦LBL 01
205 RCL 07
206 RCL 08
207 *
208 STO 00
209 .0867
210 RCL 14
211 /
212 X<>Y
213 RCL 13
214 *
215 *
216 STO 12
217 LASTX
218 *
219 RCL 13
220 SQRT
221 *
222 4.934
223 *
224 STO 11
225 RTN
```

Display

```
226♦LBL 10
227 ARCL X
228 AVIEW
229 STOP
230 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	NR	50	
	temp storage index		
	$a/T^{1/2}$		
	$(V + b)$		
	$(V - b)$		
05	P	55	
	V		
	n		
	f		
	T		
10	control	60	
	a		
	b		
	T_c		
	P_c		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 61 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Ideal	Redlich-Kwong
01		Calc a, b	
02		Inputting data	Calculate

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EQUATIONS OF STATE

PROGRAM REGISTERS NEEDED: 47

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 5)

ROW 2 (5 - 10)

ROW 3 (11 - 17)

ROW 4 (18 - 24)

ROW 5 (25 - 30)

ROW 6 (30 - 37)

ROW 7 (38 - 47)

ROW 8 (48 - 55)

ROW 9 (55 - 65)

ROW 10 (66 - 73)

ROW 11 (73 - 79)

ROW 12 (80 - 91)

ROW 13 (91 - 98)

ROW 14 (98 - 108)

ROW 15 (109 - 121)

ROW 16 (122 - 130)

ROW 17 (130 - 138)

ROW 18 (139 - 151)

ROW 19 (152 - 164)

ROW 20 (165 - 177)

ROW 21 (178 - 190)

ROW 22 (191 - 202)

ROW 23 (203 - 210)

ROW 24 (211 - 222)

ROW 25 (222 - 229)

ROW 26 (230 - 230)

REVERSIBLE POLYTROPIC PROCESS FOR AN IDEAL GAS

This program may be used to solve interchangeably between pressure ratio, volume ratio, temperature ratio, and density ratio for polytropic processes involving ideal gases. Polytropic processes are defined by the relation

$$PV^n = C.$$

Isentropic processes are special cases of polytropic processes. For isentropic processes, k, the specific heat ratio, is equal to n.

Equations:

$$\frac{P_2}{P_1} = \left(\frac{V_2}{V_1}\right)^{-n} = \left(\frac{T_2}{T_1}\right)^{\frac{n}{n-1}} = \left(\frac{\rho_2}{\rho_1}\right)^{n}$$

where

P_2/P_1 is the final pressure divided by the initial pressure;
V_2/V_1 is the final volume divided by the initial volume;
T_2/T_1 is the final temperature divided by the initial temperature;
ρ_2/ρ_1 is the final density divided by the initial density.

Note that in the display "ρ" is indicated by "D".

Example:

A compressor has a compression ratio of 8.5 (V_1/V_2). The polytropic constant is 1.43. If inlet air is at 300 K, what is the outlet temperature? What is the outlet pressure in atmospheres if the inlet pressure is one atmosphere?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] GAS [ALPHA]	N?
1.43 [R/S]	0.00
8.5 [1/x] [XEQ] [ALPHA] V2/V1 [ALPHA]	0.00
[XEQ] [ALPHA] T2/T1 [ALPHA]	T2/T1=2.51
300 [X]	752.96
0 [XEQ] [ALPHA] P2/P1 [ALPHA]	P2/P1=21.33

User Instructions

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution and input n		[XEQ] GAS	N
		n	[R/S]	
3	Input one ratio	P2/P1	[XEQ] P2/P2	0.00
		V2/V1	[XEQ] V2/V1	0.00
		T2/T1	[XEQ] T2/T1	0.00
		$\rho2/\rho1$	[XEQ] D2/D1	0.00
4	Calculate any ratio(s)	0	[XEQ] P2/P1	P2/P1 =
		0	[XEQ] V2/V1	V2/V1 =
		0	[XEQ] T2/T1	T2/T1 =
		0	[XEQ] D2/D1	D2/D1 =

Program Listings

Initialization

```
LBL "GAS"
FIX 2
"N?"
PROMPT
STO 02
1
-
STO 03
1/X
STO 04
0
STOP
```

P_2/P_1

```
LBL "P2/P1"
X=0?
GTO 01
STO 05
0
STOP
LBL 01
RCL 05
"P2/P1="
GTO d
STOP
```

V_2/V_1

```
LBL "V2/V1"
X=0?
GTO 01
RCL 02
CHS
Y↑X
STO 05
0
STOP
LBL 01
RCL 05
RCL 02
CHS
1/X
Y↑X
"V2/V1="
GTO d
```

T_2/T_1

```
LBL "T2/T1"
X=0?
GTO 01
RCL 02
RCL 04
*
Y↑X
STO 05
0
STOP
LBL 01
RCL 05
RCL 03
RCL 02
/
Y↑X
"T2/T1="
GTO d
```

ρ_2/ρ_1

```
LBL "D2/D1"
X=0?
GTO 01
RCL 02
Y↑X
STO 05
0
STOP
LBL 01
RCL 05
RCL 02
1/X
Y↑X
"D2/D1="
```

Display

```
LBL d
ARCL X
AVIEW
STOP
.END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	n		
	n-1		
	1/n-1		
05	P2/P1	55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 29 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		NONE	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

REVERSIBLE POLYTROPIC PROCESS
FOR AN IDEAL GAS
PROGRAM REGISTERS NEEDED: 24

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (5 - 13)

ROW 3 (13 - 17)

ROW 4 (18 - 22)

ROW 5 (23 - 26)

ROW 6 (27 - 37)

ROW 7 (38 - 41)

ROW 8 (41 - 46)

ROW 9 (47 - 57)

ROW 10 (57 - 59)

ROW 11 (59 - 65)

ROW 12 (65 - 72)

ROW 13 (72 - 77)

CONDUIT FLOW

This program solves for the average velocity, or the pressure drop for viscous, incompressible flow in conduits.

Equations:

$$v^2 = \frac{\Delta P/\rho}{2\left(f\,\frac{L}{D} + \frac{K_T}{4}\right)}$$

For laminar flow (Re < 2300)

$$f = 16/Re.$$

For turbulent flow (Re > 2300)

$$\frac{1}{\sqrt{f}} = 1.737\ \ln\frac{D}{\varepsilon} + 2.28 - 1.737\ \ln\left(4.67\,\frac{D}{\varepsilon Re\,\sqrt{f}} + 1\right)$$

is solved by Newton's method.

$$\frac{1}{\sqrt{f_0}} = 1.737\ \ln\frac{D}{\varepsilon} + 2.28$$

is used an an initial guess in the iteration.

where: Re is the Reynolds number, defined as $\rho Dv/\mu$;

- D is the pipe diameter;
- ε is the dimension of irregularities in the conduit surface (see table 2);
- f is the Fanning friction factor for conduit flow;
- ΔP is the pressure drop along the conduit;
- ρ is the density of the fluid;
- μ is the viscosity of the fluid;
- ν is the kinematic viscosity of the fluid and $\mu=\rho\nu$;
- L is the conduit length;
- v is the average fluid velocity;
- K_T is the total of the applicable fitting coefficients in table 1.

Table 1

Fitting Coefficients

Fitting	K
Globe valve, wide open	7.5—10
Angle valve, wide open	3.8
Gate valve, wide open	0.15—0.19
Gate valve, 3/4 open	0.85
Gate valve, 1/2 open	4.4
Gate valve, 1/4 open	20
90° elbow	0.4—0.9
Standard 45° elbow	0.35—0.42
Tee, through side outlet	1.5
Tee, straight through	.4
180° bend	1.6
Entrance to circular pipe	0.25—0.50
Sudden expansion	$(1-A_{up}/A_{dn})^2$*
Acceleration from $v = 0$ to $v = v_{entrance}$	1.0

*A_{up} is the upstream area and A_{dn} is the downstream area.

Table 2

Surface Irregularities

Material	ε(feet)	ε(meters)
Drawn or Smooth Tubing	5.0×10^{-6}	1.5×10^{-6}
Commercial Steel or Wrought Iron	1.5×10^{-4}	4.6×10^{-5}
Asphalted Cast Iron	4.0×10^{-4}	1.2×10^{-4}
Galvanized Iron	5.0×10^{-4}	1.5×10^{-4}
Cast Iron	8.3×10^{-4}	2.5×10^{-4}
Wood Stave	6.0×10^{-4} to 3.0×10^{-3}	1.8×10^{-4} to 9.1×10^{-4}
Concrete	1.0×10^{-3} to 1.0×10^{-2}	3.0×10^{-4} to 3.0×10^{-3}
Riveted Steel	3.0×10^{-3} to 3.0×10^{-2}	9.1×10^{-4} to 9.1×10^{-3}

Reference:

Welty, Wicks, Wilson, *Fundamentals of Momentum, Heat and Mass Transfer*, John Wiley and Sons, Inc., 1969.

Remarks:

The correlation gives meaningless results in the region $2300 < Re < 4000$.

The solution requires an iterative procedure. The time for solution will range from 10 seconds for ΔP, to several minutes for v. The display setting is used to determine when the solution for v is adequately accurate. Time for solution of v is roughly proportional to the number or significant digits in the display setting.

If the conduit is not circular, an equivalent diameter may be calculated using the formula below:

$$D_{eq} = 4\frac{\text{cross sectional area}}{\text{wetted perimeter}}$$

Unitary consistency must be maintained.

Example:

A heat exchanger has 20, 3 meter tube passes (60 m of pipe) with 180 degrees bends connecting each pair of tubes (from table 1, $K_T = 10 \times 1.6$). The fluid is water ($\nu = 9.3 \times 10^{-7} m^2/s$, $\rho = 10^3 kg/m^3$). The surface roughness is 3×10^{-4}m and the diameter is 2.54×10^{-2}m. If the fluid velocity is 3.05 m/s, what is the pressure loss? What is the Reynolds number? What is the Fanning friction factor?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[///] [ENG] 3	
[XEQ] [ALPHA] CONDUIT [ALPHA]	U=?
9.3 [EEX] [CHS] 7 [ENTER↑]	
[EEX] 3 [X] [R/S]	RHO=?
[EEX] 3 [R/S]	E=?
3 [EEX] [CHS] 4 [R/S]	L=?
60 [R/S]	D=?
2.54 [EEX] [CHS] 2 [R/S]	KT=?
16 [R/S]	V=?
3.05 [R/S]	DP=?
[R/S]	DP=521.9E3
[R/S]	Re=83.30E3
[R/S]	F=10.18E-3

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize		[XEQ] CONDUIT	U=?
3	Key in viscosity	μ	[R/S]	RHO=?
4	Key in density	ρ	[R/S]	E=?
5	Key in surface irregularity	ε	[R/S]	L=?
6	Key in length of conduit	L	[R/S]	D=?
7	Key in equivalent diameter	D	[R/S]	KT=?
8	Key in total fitting coefficient	K_T	[R/S]	V=?
9	Key in velocity if known, otherwise			
	press [R/S]	v	[R/S]	DP=?
10	Key in pressure drop if known, otherwise			
	press [R/S]. (either v or ΔP must be			
	known)	ΔP	[R/S]	(unknown)=
11	Optional: obtain Reynolds no.		[R/S]	Re=
	Optional: obtain Fanning friction factor		[R/S]	F=
12	To change just v or ΔP		[XEQ] C	V=?
13	Go to step 9			
14	To begin a whole new problem go to step 2			

Program Listings

Input

```
01♦LBL "CON
DUIT"
02 SF 21
03 SF 27
04 "U=?"
05 PROMPT
06 STO 09
07 "RHO=?"
08 PROMPT
09 STO 10
10 ST/ 09
11 "E=?"
12 PROMPT
13 STO 14
14 "L=?"
15 PROMPT
16 STO 03
17 "D=?"
18 PROMPT
19 STO 13
20 "KT=?"
21 PROMPT
22 4
23 /
24 STO 08
25♦LBL C
26 CF 22
27 "V=?"
28 PROMPT
29 SF 00
30 FS? 22
31 CF 00
32 STO 02
33 "DP=?"
34 PROMPT
```

1st V

Calculate ΔP

```
35 STO 04
36 XEQ 09
37 FS? 00
38 GTO 03
39 RCL 02
40 X↑2
41 *
42 RCL 10
43 *
44 STO 04
45 "DP="
46 GTO 10
```

Iterate to find V using 1st V as guess

```
47♦LBL 03
48 RND
49 STO 00
50 XEQ 08
51 RND
52 RCL 00
53 X<>Y
54 X≠Y?
55 GTO 03
56 "V=?"
57 RCL 02
58 GTO 10
```

Calculate constants

```
59♦LBL 09
60 RCL 10
61 RCL 13
62 RCL 14
63 /
64 STO 06
65 LN
66 1.737
67 STO 07
68 *
69 2.28
70 +
71 STO 12
72 STO 05
73 FS? 00
74 GTO 07
```

Is flow turbulent?

```
75♦LBL 08
76 16
77 RCL 02
78 RCL 13
79 *
80 RCL 09
81 /
82 STO 01
83 2300
84 X<=Y?
85 GTO 02
86 RDN
87 /
88 SQRT
89 1/X
90 STO 05
91 GTO 07
```

Iterate to find $\frac{1}{\sqrt{f}}$

```
92♦LBL 02
93 RCL 12
94 RCL 05
95 -
96 4.67
97 RCL 06
98 *
99 RCL 01
```

Program Listings

```
100 /
101 RCL 05
102 *
103 1
104 +
105 STO 11
106 LN
107 RCL 07
108 *
109 -
110 RCL 11
111 1/X
112 CHS
113 1
114 +
115 RCL 07
116 *
117 RCL 05
118 /
119 1
120 +
121 /
122 ST+ 05
123 RCL 05
124 /
125 ABS
126 1 E-3
127 X<=Y?
128 GTO 02
129♦LBL 07
130 RCL 05
131 1/X
132 X↑2
133 RCL 03
134 *
135 RCL 13
136 /
137 RCL 08
138 +
139 2
140 *
141 RCL 04
142 RCL 10
143 /
144 X<>Y
145 FS? 00
146 GTO 00
147 RTN
148♦LBL 00
149 /
150 SQRT
151 STO 02
152 RTN
153♦LBL 10
154 ARCL X
155 AVIEW
156 "Re="
157 ARCL 01
158 AVIEW
159 "F="
160 RCL 05
161 1/X
162 X↑2
163 ARCL X
164 AVIEW
165 END
```

Output

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	V	50	
	Re		
	V		
	L		
	ΔP		
05	$1/\sqrt{f}$	55	
	ρ/ε		
	1.737		
	$K_T/4$		
	μ		
10	ρ	60	
	used		
	$1/\sqrt{f_0}$		
	D		
	ε		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 52 USER MODE

ENG ___ FIX ___ SCI ___ ON x OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		calculate V	calculate ΔP
22		calculate ΔP	calculate V
21	S	printer enable	printer disable
27	S	user mode on	user mode off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
change v or ΔP	C		

CONDUIT FLOW

PROGRAM REGISTERS NEEDED: 38

HEWLETT PACKARD
SOLUTIONS BOOK
CHEMICAL ENGINEERING

ROW 1 (1 : 2)

ROW 2 (3 : 7)

ROW 3 (7 : 14)

ROW 4 (14 : 20)

ROW 5 (20 : 27)

ROW 6 (27 : 33)

ROW 7 (33 : 41)

ROW 8 (42 : 50)

ROW 9 (50 : 57)

ROW 10 (58 : 66)

ROW 11 (66 : 74)

ROW 12 (74 : 83)

ROW 13 (83 : 93)

ROW 14 (94 : 103)

ROW 15 (104 : 116)

ROW 16 (117 : 126)

ROW 17 (126 : 137)

ROW 18 (138 : 148)

ROW 19 (149 : 157)

ROW 20 (157 : 165)

ROW 21 (165 : 165)

FLUID TRANSPORT NUMBERS

It is common practice in the fields of heat, mass, and momentum transfer to lump the many variables involved into dimensionless groups. These dimensionless groups, or fluid transport numbers, greatly simplify correlating experimental data and handling calculations once correlations have been obtained.

Before you start to solve a problem, pick a consistent unit system. For instance, make the units of length, feet; the units of temperature, degrees Fahrenheit; the units of time, hours; and the units of energy, British thermal units. Once you have a unit system in mind, convert all of your variables to that system before use. To calculate the Nusselt number using the system just outlined, the inputs would have to be in the following units:

h:Btu/°F-hr-ft^2 x:ft k:Btu/°F-hr-ft.

The dimensionless groups provided in these programs are Reynolds number Re, Nusselt number Nu, Nusselt number for mass transfer Nu_{ab}, Lewis number Le, Schmidt number Sc, Stanton number St and Prandtl number Pr. All of these numbers are correlated using interchangeable solutions.

Table of Equations

Number	Symbol	Formula	Use
Reynolds	Re	$\rho\ x\ v/\mu$	Momentum, mass and heat transfer where velocity and viscosity must be considered
Nusselt-heat	Nu	$h\ x/k$	Convective heat transfer
Biot	Bi	$h\ x/k$	Combinations of convective and conductive transport systems
Nusselt-mass	Nu_{ab}	$k_c\ x/D_{ab}$	Convective mass transfer
Stanton	St	$h/\rho\ v\ c_p$	Convective heat transfer
Lewis	Le	$k_c/\rho\ c_p D_{ab}$	Convective mass transfer
Schmidt	S_c	$\mu/\rho\ D_{ab}$	Convective mass transfer
Prandtl	Pr	$\mu\ c_p/k$	Convective heat transfer

where:

ρ = density ("D" is used in the display)

v = velocity

μ = viscosity

x = significant dimension

h = convective heat transfer coefficient

k = conductive heat transfer coefficient

k_c = convective mass transfer coefficient

D_{ab} = mass diffusivity

c_p = constant pressure heat capacity

Example:

At 60°F, the properties of water are:

ρ = 62.3 lb/ft^3

c_p = 1.00 Btu/lb°F

μ = 0.760 x 10^{-3} lb/ft sec

ν = 1.22 x 10^{-5} ft^2/sec

k = 0.340 Btu/hr-ft-°F

Assume that fluid velocity is 37 feet per second and that the critical dimension is 6 inches. Calculate the Reynolds number using viscosity μ and density ρ

Load "Reynolds number" program (#1)

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] RE [ALPHA]	X
6 [ENTER] 12 [÷] [R/S]	V
37 [R/S]	D
62.3 [R/S]	U
0.76 [EEX] 3 [CHS] [R/S]	Re
[R/S]	Re=1.517E6

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution (input 4, and skip			
	unknown with [R/S])		[XEQ] Re	X
	significant dimension	x	[R/S]	V
	fluid velocity	v	[R/S]	D
	fluid density	ρ	[R/S]	U
	fluid viscosity	μ	[R/S]	Re
	Reynolds number	Re	[R/S]	(unknown)=

Program Listings

Initialization

```
01◆LBL "Re"
02 ENG 3
03 CF 00
```

Input

```
04 CLX
05 "X"
06 PROMPT
07 X=0?
08 XEQ b
09 STO 08
10 CLX
11 "V"
12 PROMPT
13 X=0?
14 XEQ b
15 STO 07
16 CLX
17 "D"
18 PROMPT
19 X=0?
20 XEQ b
21 STO 04
22 CLX
23 "U"
24 PROMPT
25 X=0?
26 XEQ a
27 STO 05
28 CLX
29 "Re"
30 PROMPT
31 X=0?
32 XEQ a
33 STO 01
```

$\frac{Re\mu}{xv\rho} = 1.0$

```
34 RCL 05
35 *
36 RCL 08
37 RCL 07
38 *
39 RCL 04
40 *
41 /
42 FC? 00
43 1/X
```

Display

```
44 CLA
45 ARCL 00
46 "⊢="
47 ARCL X
48 AVIEW
49 STOP
```

```
50◆LBL b
51 SF 00
```

Label output rtn with 1.0

```
52◆LBL a
53 ASTO 00
54 1
55 RTN
56 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	label for output	50	
	Re		
	ρ		
05		55	
	μ		
	v		
	x		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 21 USER MODE
ENG 3 FIX ___ SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		INVERT	NO-INVERT

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	For Nusselt-heat		[XEQ] NU	H
3	Go to step 5			
4	For Biot		[XEQ] BI	H
5	Key in 3 of the following 4 inputs.			
	Key [R/S] to skip the unknown			
	convective heat transfer coefficient	h	[R/S]	X
	significant dimension	x	[R/S]	K
	conductive heat transfer coefficient	k	[R/S]	NU (or) BI
	Nusselt number or Biot number	Nu or Bi	[R/S]	(unknown)=
6	For Nusselt-mass		[XEQ] NUab	Kc
7	Key in 3 of the following 4 inputs.			
	Key [R/S] to skip the unkown			
	convective mass transfer coefficient	k_c	[R/S]	X
	significant dimension	x	[R/S]	Dab
	mass diffusivity	D_{ab}	[R/S]	NUab
	Nusselt number	Nu_{ab}	[R/S]	(unknown)=

Program Listings

Initialization

```
01♦LBL "NU"
02 "NU"
03 SF 01
04 GTO 01
05♦LBL "BI"
06 "BI"
07 SF 01
08 GTO 01
09♦LBL "NUa
b"
10 "NUab"
11 CF 01
12♦LBL 01
13 ASTO 01
14 SF 00
```

Input

```
15 ENG 3
16 CLX
17 "Kc"
18 FS? 01
19 "H"
20 PROMPT
21 X=0?
22 XEQ a
23 STO 05
24 CLX
25 "X"
26 PROMPT
27 X=0?
28 XEQ a
29 STO 08
30 CLX
31 "Dab"
32 FS? 01
33 "K"
34 PROMPT
35 X=0?
36 XEQ b
37 STO 06
38 CLX
39 CLA
40 ARCL 01
41 PROMPT
42 X=0?
43 XEQ b
44 STO 02
```

Calculation

```
45 RCL 06
46 *
47 RCL 08
48 /
49 RCL 05
50 /
51 FS? 00
52 1/X
```

Display

```
53 CLA
54 ARCL 00
55 "⊢="
56 ARCL X
57 AVIEW
58 STOP
```

Label for unknown

```
59♦LBL a
60 CF 00
61♦LBL b
62 ASTO 00
63 1
64 RTN
65 .END.
```

70

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	label for unknown	50	
	"NU","BI",or "NUab"		
	Nu (Bi)		
05	h(kc)	55	
	k(Dab)		
	x		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 27 USER MODE

ENG 3 FIX ____ SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		INVERT	NO-INVERT
01		Nu or Bi	Nu_{ab}

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	For Stanton (input 4, and skip unknown			
	with [R/S])		[XEQ] ST	H
	convective heat transfer coefficient	h	[R/S]	D
	fluid density	ρ	[R/S]	V
	fluid velocity	v	[R/S]	CP
	fluid heat capacity	c_p	[R/S]	ST
	Stanton number	St	[R/S]	(unknown)=
3	For Lewis (input 4, and skip unknown			
	with [R/S])		[XEQ] Le	Kc
	convective mass transfer coefficient	k_c	[R/S]	D
	fluid density	ρ	[R/S]	Dab
	mass diffusivity	D_{ab}	[R/S]	CP
	fluid heat capacity	c_p	[R/S]	Le
	Lewis number	Le	[R/S]	(unknown)=

Program Listings

01◆LBL "Le" 02 SF 01 03 GTO 01 04◆LBL "ST" 05 CF 01 06◆LBL 01 07 SF 00 08 ENG 3	Initialization
09 CLX 10 "H" 11 FS? 01 12 "Kc" 13 PROMPT 14 X=0? 15 XEQ a 16 STO 05 17 CLX 18 "D" 19 PROMPT 20 X=0? 21 XEQ b 22 STO 04 23 CLX 24 "V" 25 FS? 01 26 "Dab" 27 PROMPT 28 X=0? 29 XEQ b 30 STO 07 31 CLX 32 "CP" 33 PROMPT 34 X=0? 35 XEQ b 36 STO 06 37 CLX 38 "ST" 39 FS? 01 40 "Le" 41 PROMPT 42 X=0? 43 XEQ b 44 STO 02	Input
45 RCL 04 46 * 47 RCL 07 48 * 49 RCL 06 50 * 51 RCL 05 52 / 53 FS? 00 54 1/X 55 CLA	Calculation
56 ARCL 00 57 "⊢=" 58 ARCL X 59 AVIEW 60 STOP 61◆LBL a 62 CF 00	Display
63◆LBL b 64 ASTO 00 65 1 66 RTN 67 .END.	Label output rtn with 1.0

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	label for unknown	50	
	St (Le)		
	P		
05	$h(k_c)$	55	
	c_p		
	v(Dab)		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 25 USER MODE

ENG 3 FIX ___ SCI ___ ON ___ OFF X

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		INVERT	NON-INVERT
01		LEWIS	STANTON

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	For Prandtl (input 3, and skip unknown			
	with [R/S])		[XEQ] PR	U
	fluid viscosity	μ	[R/S]	CP
	fluid heat capacity	c_p	[R/S]	K
	conductive heat transfer coefficient	k	[R/S]	PR
	Prandtl number	Pr	[R/S]	(unknown)=
3	For Schmidt (input 3, and skip unknown			
	with [R/S])		[XEQ] SC	U
	fluid viscosity	μ	[R/S]	D
	fluid density	ρ	[R/S]	Dab
	mass diffusivity	D_{ab}	[R/S]	Sc
	Schmidt number	S_c	[R/S]	(unknown)=

Program Listings

Initialization

```
01♦LBL "PR"
02 CF 01
03 GTO 01
04♦LBL "SC"
05 SF 01
06♦LBL 01
07 SF 00
08 ENG 3
```

Input

```
09 CLX
10 "U"
11 PROMPT
12 X=0?
13 XEQ b
14 STO 07
15 CLX
16 "CP"
17 FS? 01
18 "D"
19 PROMPT
20 X=0?
21 XEQ b
22 STO 04
23 CLX
24 "K"
25 FS? 01
26 "Dab"
27 PROMPT
28 X=0?
29 XEQ a
30 STO 06
31 CLX
32 "PR"
33 FS? 01
34 "SC"
35 PROMPT
36 X=0?
37 XEQ a
38 STO 03
```

Calculate

```
39 RCL 06
40 *
41 RCL 04
42 RCL 07
43 *
44 /
45 FS? 00
46 1/X
```

Display

```
47 CLA
48 ARCL 00
49 "⊢="
50 ARCL X
51 AVIEW
52 STOP
```

Label output rtn with 1.0

```
53♦LBL a
54 CF 00
55♦LBL b
56 ASTO 00
57 1
58 RTN
59 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	label for output	50	
	Pr (Sc)		
	cp		
05		55	
	K(Dab)		
	μ		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 23 USER MODE

ENG 3 FIX ____ SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		INVERT	NON-INVERT
01		Schmidt	Prandtl

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FLUID TRANSPORT NUMBERS
REYNOLDS NUMBER
PROGRAM REGISTERS NEEDED: 13

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 5)

ROW 2 (6 - 14)

ROW 3 (14 - 23)

ROW 4 (23 - 31)

ROW 5 (32 - 42)

ROW 6 (42 - 50)

ROW 7 (50 - 56)

FLUID TRANSPORT NUMBERS
NUSSELT AND BIOT NUMBERS
PROGRAM REGISTERS NEEDED: 19

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (5 - 8)

ROW 3 (9 - 10)

ROW 4 (11 - 17)

ROW 5 (18 - 25)

ROW 6 (26 - 32)

ROW 7 (33 - 41)

ROW 8 (42 - 51)

ROW 9 (52 - 59)

ROW 10 (60 - 65)

FLUID TRANSPORT NUMBERS
STANTON AND LEWIS NUMBERS
PROGRAM REGISTERS NEEDED: 18

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (4 - 10)

ROW 3 (11 - 18)

ROW 4 (18 - 26)

ROW 5 (26 - 32)

ROW 6 (33 - 40)

ROW 7 (40 - 49)

ROW 8 (50 - 58)

ROW 9 (58 - 66)

ROW 10 (67 - 67)

FLUID TRANSPORT NUMBERS
SCHMIDT AND PRANDTL NUMBERS
PROGRAM REGISTERS NEEDED: 16

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (4 - 10)

ROW 3 (11 - 18)

ROW 4 (18 - 26)

ROW 5 (26 - 32)

ROW 6 (33 - 40)

ROW 7 (41 - 49)

ROW 8 (50 - 57)

ROW 9 (58 - 59)

SINGLE STAGE EQUILIBRIUM FLASH CALCULATION

THIS PROGRAM REQUIRES 1 ADDITIONAL MEMORY MODULE

Given the number of components (n), the mole fraction (z_i), the equilibrium ratio between phases (K_i) for each component in the feed and a first guess for the mole ratio (V/F), the program will use a second order Newton convergence scheme to converge the Rachford-Rice equation to nearly zero (10^{-6}).

$$f(V/F) = \sum_{i=1}^{n} z_i(K_i-1) / [(K_i-1)(V/F) + 1] = 0, \text{ Rachford-Rice Equation}$$

Newton Convergence Method:

$$(V/F)_{i+1} = (V/F)_i - f[(V/F)_i] / f'[(V/F)_i]$$

$$f'[(V/F)_i] = -\sum_{i=1}^{n} z_i(K_i-1)^2 / [(K_i-1)(V/F) + 1]^2$$

Next the program can solve for (V/L) and the final compositions of the liquid and the vapor by: (V/L) = (V/F) / [1 - (V/F)]

$$x_i = z_i [1 + (V/L)] / [1 + K_i(V/L)]$$

$$y_i = z_i [1 + (L/V)] / [1 + (1/K_i)(L/V)]$$

Note: Maximum number of components = 10



A five component mixture of hydrocarbons is fed to a steady-state flash vaporization giving product equilibrium at 270°F and 50 psia. The details are given below:

Component	K_i @ 270°F & 50 psia	z_i
C_3	12.75	0.15
C_4	5.61	0.25
C_6	1.40	0.05
C_7	0.705	0.30
C_8	0.375	0.25

Calculate (V/F) and the composition of both the vapor and the liquid product streams.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 025	
[XEQ] [ALPHA] SSE [ALPHA]	N?
5 [R/S]	K?
12.75 [R/S]	Z?
.15 [R/S]	K?
5.61 [R/S]	Z?
.25 [R/S]	K?
1.4 [R/S]	Z?
.05 [R/S]	K?
.705 [R/S]	Z?
.3 [R/S]	K?
.375 [R/S]	Z?
.25 [R/S]	?
.5 [R/S]	V/F=0.80243
[R/S]	X=0.01438
[R/S]	Y=0.18339

Keystrokes:	Display:
[R/S]	X=0.05320
[R/S]	Y=0.29845
[R/S]	X=0.03785
[R/S]	Y=0.05299
[R/S]	X=0.39304
[R/S]	Y=0.27709
[R/S]	X=0.50153
[R/S]	Y=0.18807

User Instructions

SIZE: 025

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize		[XEQ] SSE	N?
3	Key in number of components	n	[R/S]	K?
4	Key in the equilibrium ratio between			
	phases of component 1	K_1	[R/S]	Z?
5	Key in the mole fraction of component 1	z_1	[R/S]	K? or ?
	(Repeat steps 4 and 5 for each component.			
	When done see "?".)			
6	Input mole ratio estimate	$(V/F)_{est}$	[R/S]	V/F=
7	Optional: compute the mole fraction of			
	each component in the liquid (x_i) and the			
	vapor (y_i)		[R/S]	X=
	(Repeat for each component.)		[R/S]	Y=
8	Optional: compute the mole ratio (V/L)		[XEQ] V/L	V/L=
9	Optional: calculate the vapor mole rate			
	and liquid mole rate		[XEQ] VL	F?
	Key in the mole feed rate	F	[R/S]	V=
			[R/S]	L=

Program Listings

Initialization

```
 01♦LBL "SSE
"
 02 CF 02
 03 FIX 5
 04 "N?"
 05 PROMPT
 06 1
 07 -
 08 STO 21
 09 STO 23
 10♦LBL A
 11 "K?"
 12 PROMPT
 13 STO IND
21
 14 RCL 21
 15 10
 16 +
 17 "Z?"
 18 PROMPT
 19 STO IND
Y
 20 FS? 02
 21 GTO C
 22 DSE 21
 23 GTO A
 24 SF 02
 25 GTO A
```

Prepare for calculation of f(V/F) or f'(V/F)

```
 26♦LBL c
 27 STO 20
 28 RCL 23
 29 STO 21
 30 0
 31 STO 24
```

Calculate f(V/F)

$$=\sum_{i=1}^{n} \frac{Zi(Ki-1)}{(Ki-1)(V/F)+1}$$

```
 32♦LBL 06
 33 XEQ 01
 34 DSE 21
 35 GTO 06
 36 XEQ 01
 37 RTN
 38♦LBL 01
 39 RCL 20
 40 RCL IND
21
 41 1
 42 -
 43 *
 44 1
 45 +
 46 FS? 00
 47 X↑2
 48 RCL IND
21
 49 RCL 21
 50 10
 51 +
 52 RDN
 53 RCL IND
T
 54 X<>Y
 55 1
 56 -
 57 FS? 00
 58 X↑2
 59 *
 60 X<>Y
 61 /
 62 RCL 24
 63 +
 64 STO 24
 65 RTN
```

set flag 00 so that subroutine c will calc. f'(V/F)

```
 66♦LBL 02
 67 RCL 20
 68 SF 00
 69 XEQ c
 70 CHS
 71 CF 00
 72 RTN
```

$0 \leq (V/F) \leq 1$

```
 73♦LBL C
 74 "?"
 75 FS?C 02
 76 PROMPT
 77 CF 01
 78 STO 20
 79 ABS
 80 FRC
 81 XEQ c
 82 STO 22
 83 ABS
```

$|f(V/F)| < 10^{-6}$

```
 84 1 E-6
 85 X>Y?
 86 GTO 04
```

Adjust by Newtons method

```
 87 XEQ 02
 88 RCL 22
 89 X<>Y
 90 /
 91 RCL 20
 92 X<>Y
 93 -
 94 GTO C
```

Display V/F

```
 95♦LBL 04
 96 RCL 20
```

Program Listings

```
"V/F="
XEQ d
```

Compute the value of Xi then Yi for L = 1 to n

```
◆LBL D
CF 01
XEQ E
RCL 23
STO 21
◆LBL 07
XEQ 05
DSE 21
GTO 07
XEQ 05
RTN
◆LBL 05
RCL IND 21
RCL 24
*
1
+
RCL 24
1
+
X<>Y
/
RCL 21
10
+
RCL IND X
X<>Y
RDN
*
"X="
XEQ d
RCL IND T
1
RCL 24
1/X
+
*
RCL 24
1/X
RCL IND 21
/
1
+
/
"Y="
XEQ d
RTN
```

Convert V/F to V/L

```
◆LBL "V/L"
SF 01
◆LBL E
RCL 20
1
RCL 20
-
/
STO 24
FC? 01
RTN
"V/L="
GTO d
```

Compute V and L

```
◆LBL "VL"
"F?"
PROMPT
ENTER↑
ENTER↑
RCL 20
*
"V="
XEQ d
-
"L="
```

Display

```
◆LBL d
ARCL X
AVIEW
STOP
RTN
.END.
```

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Kn	50	
	Kn-1		
	Kn-2		
	Kn-3		
	Kn-4		
05	Kn-5	55	
	Kn-6		
	Kn-7		
	Kn-8		
	Kn-9		
10	Zn	60	
	Zn-1		
	Zn-2		
	Zn-3		
	Zn-4		
15	Zn-5	65	
	Zn-6		
	Zn-7		
	Zn-8		
	Zn-9		
20	Trial (V/F)	70	
	temp storage (INP Operations)		
	f(V/F)		
	n-1		
	temp storage		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 025 TOT. REG. 70 USER MODE
ENG ___ FIX 5 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		f'(V/F)	f(V/F)
01		RETURN	display

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SINGLE STAGE EQUILIBRIUM
FLASH CALCULATION
PROGRAM REGISTERS NEEDED: 46

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (4 - 11)

ROW 3 (12 - 19)

ROW 4 (19 - 24)

ROW 5 (25 - 31)

ROW 6 (31 - 37)

ROW 7 (38 - 47)

ROW 8 (48 - 56)

ROW 9 (57 - 66)

ROW 10 (67 - 73)

ROW 11 (74 - 81)

ROW 12 (81 - 87)

ROW 13 (87 - 95)

ROW 14 (96 - 100)

ROW 15 (100 - 106)

ROW 16 (106 - 113)

ROW 17 (114 - 123)

ROW 18 (124 - 130)

ROW 19 (131 – 140)

ROW 20 (141 – 146)

ROW 21 (146 – 152)

ROW 22 (153 – 158)

ROW 23 (158 – 163)

ROW 24 (164 – 169)

ROW 25 (170 – 175)

WEAK ACID/BASE TITRATION CURVE

THIS PROGRAM REQUIRES 1 ADDITIONAL MEMORY MODULE

This program concerns the titration of weak acids H_4A, H_3A, H_2A and HA, and weak bases $A(OH)_4$, $A(OH)_3$, $A(OH)_2$ and AOH. Given the dissociation constants K_n, and the beginning volume and molarity (V and M) of the acid or base being titrated, and the normality (N) of the titrant, the pH of the solution can be calculated for different volumes of titrant added. Since an iterative technique is being used, an initial guess for pH must be supplied. If it is a base that is being titrated then the main routine calculates pOH; however, pOH is converted to pH for display.

References:

Breneman, G.L., "A General Acid-Base Titration Curve Computer Program," Journal of Chemical Education; vol. 51, pp. 812-813.

Example:

For phosphoric acid, H_3PO_4, $K_1 = 7.5 \times 10^{-3}$, $K_2 = 6.2 \times 10^{-8}$ and $K_3 = 1 \times 10^{-12}$. Plot a titration curve from 0 to 75 ml of 0.500 N NaOH added to 50.00 ml of 0.200 M H_2PO_4.

Solution(s): For a complete solution for purposes of a plot, calculations of the pH at approximately 45 different titrant volumes are required. For purposes of illustration the following 15 calculations will suffice:

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 026	
[XEQ[[ALPHA] A/B [ALPHA]	ACID Y/N
Y [R/S]	K?
7.5 [EEX] [CHS] 3 [R/S]	K?
6.2 [EEX] [CHS] 8 [R/S]	K?
[EEX] [CHS] 12 [R/S]	K?
[A]	V?
50 [R/S]	M?
.2 [R/S]	N?
.5 [R/S]	INIT. GUESS
[R/S]	VOLUME ADDED
0 [R/S]	PH=1.454

Keystrokes:	Display:
[R/S]	VOLUME ADDED
10 [R/S]	PH=2.192
[R/S]	VOLUME ADDED
15 [R/S]	PH=2.637
[R/S]	VOLUME ADDED
19.4 [R/S]	PH=3.654
[R/S]	VOLUME ADDED
20 [R/S]	PH=4.677
[R/S]	VOLUME ADDED
20.4 [R/S]	PH=5.527
[R/S]	VOLUME ADDED
21 [R/S]	PH=5.930
[R/S]	VOLUME ADDED
25 [R/S]	PH=6.731
[R/S]	VOLUME ADDED
35 [R/S]	PH=7.685
[R/S]	VOLUME ADDED
39 [R/S]	PH=8.483
[R/S]	VOLUME ADDED
39.8 [R/S]	PH=9.142
[R/S]	VOLUME ADDED
40.1 [R/S]	PH=9.832
[R/S]	VOLUME ADDED
40.8 [R/S]	PH=10.584
[R/S]	VOLUME ADDED
45 [R/S]	PH=11.457
[R/S]	VOLUME ADDED
75 [R/S]	PH=12.845

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode		[USER]	
2	Begin execution:		[XEQ] A/B	ACID Y/N
	if acid	Y	[R/S]	K?
	if base	N	[R/S]	K?
3	Input K_1, K_2, K_3 and K_4 as needed	K_1	[R/S]	K?
		K_2	[R/S]	K?
		K_3	[R/S]	K?
		K_4	[R/S]	K?
4	When all K's are input, continue		[A]	V?
		V	[R/S]	M?
		<u>M</u>	[R/S]	N?
		<u>N</u>	[R/S]	INIT. GUESS
5	Begin calculations:			
	input guess for pH*	pHest	[R/S]	VOLUME ADDED
	input volume of titrant added	V	[R/S]	PH =
6	Repeat step 5			
*	A guess for pH need not be entered IF:			
	1) The volume of titrate is going to be			
	zero			
	OR			
	2) The pH in the display left over from			
	the last problem is suitable.			

Program Listings

```
01♦LBL "A/B"
02 FIX 3
03 "ACID Y/N"
04 AON
05 PROMPT
06 AOFF
07 ASTO X
08 "Y"
09 ASTO Y
10 CF 00
11 X=Y?
12 SF 00
13 CLRG
14♦LBL E
15 1
16 ST+ 25
17 "K?"
18 PROMPT
19 STO IND 25
20 GTO E
21♦LBL A
22 "V?"
23 PROMPT
24 STO 23
25 "M?"
26 PROMPT
27 *
28 STO 24
29 "N?"
30 PROMPT
31 STO 21
32 RCL 25
33 1
34 +
35 STO 00
36 "INIT. GUESS"
37 PROMPT
```

Initialization

```
38♦LBL B
39 "VOLUME ADDED"
40 PROMPT
41 CF 02
42 X=0?
43 SF 02
44 RCL 21
45 X<>Y
46 *
47 LASTX
48 RCL 23
49 +
50 /
51 STO 20
```

Initialize for coefficients

```
52 RCL 24
53 LASTX
54 /
55 STO 22
56 R↑
57 STO 08
58 RCL 00
59 STO 07
```

$[Na^+] \rightarrow R_{20}$

```
60♦LBL c
```

$c \rightarrow R_{22}$

$pH_{est} \rightarrow R_{08}$

```
61 RCL 07
62 .5
63 +
64 STO 25
65 ISG 25
66 RCL IND 25
67 RCL 20
68 +
69 .5
70 ST- 25
71 RDN
72 DSE 25
73 RCL 25
74 RCL 22
75 *
76 -
77 RCL IND 25
78 *
79  E-14
80 -
81♦LBL b
82 DSE 25
83 GTO 00
84 RCL 07
85 X<> 25
86 X<>Y
87 RCL 25
88 10
89 +
90 RDN
91 STO IND T
92 DSE 07
93 GTO c
94 RCL 00
```

Calculate coefficients

Program Listings

```
 95 STO 25
 96 RCL 01
 97 RCL 20
 98 +
 99 STO 10
100 RCL 01
101 RCL 22
102 *
103 SQRT
104 FS?C 02
105 GTO 01
106 GTO B
107♦LBL 00
108 RCL IND
25
109 *
110 GTO b
--------------- Change pH est by ± 2
111♦LBL a
112 2
113 FS? 00
114 CHS
115 ST+ 08
116♦LBL B
117 RCL 08
118 XEQ d
119 CHS
120 10↑X
121♦LBL 01
122 ENTER↑
123 ENTER↑
124 GTO IND
25
--------------- start forming f'([H+])
125♦LBL 05
126 6
127 *
128 RCL 10
129 5
130 XEQ 06
131 RCL 11
132 4
133 XEQ 06
134 RCL 12
135 3
136 XEQ 06
137 RCL 13
138 2
139 XEQ 06
140 RCL 14
141 XEQ 08
142 RCL 13
143 XEQ 07
144 RCL 14
145 XEQ 07
146 RCL 15
147♦LBL 09
148 +
149 RCL 16
150 /
151 -
152 X<0?
153 GTO a
154 STO 09
155 %CH
156 ABS
157 1
158 X<=Y?
159 SF 02
160 RCL 09
161 FS?C 02
162 GTO 01
163 LOG
164 CHS
--------------- Display
165 XEQ d
166 "PH="
167 ARCL X
168 AVIEW
169 STOP
170 GTO B
--------------- Convert to base
171♦LBL d
172 FS? 00
173 RTN
174 14
175 X<>Y
176 -
177 RTN
--------------- Common routine
178♦LBL 06
179 *
180♦LBL 07
181 +
182 *
183 RTN
184♦LBL 08
185 +
186 STO 16
187 CLX
188 RCL 10
189 XEQ 07
190 RCL 11
191 XEQ 07
192 RCL 12
193 FS?C 02
194 GTO 09
```

Program Listings

```
195 GTO 07
196♦LBL 04
197 5
198 *
199 RCL 10
200 4
201 XEQ 06
202 RCL 11
203 3
204 XEQ 06
205 RCL 12
206 2
207 XEQ 06
208 RCL 13
209 XEQ 08
210 RCL 13
211 XEQ 07
212 RCL 14
213 GTO 09
214♦LBL 03
215 4
216 *
217 RCL 10
218 3
219 XEQ 06
220 RCL 11
221 2
222 XEQ 06
223 RCL 12
224 XEQ 08
225 RCL 13
226 GTO 09
227♦LBL 02
228 3
229 *
230 RCL 10
231 2
232 XEQ 06
233 RCL 11
234 SF 02
235 GTO 08
236 END
```

$f'[(H^+)]$

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	#K's +1	50	
	K1		
	K2		
	K3		
	K4		
05		55	
	Counter		
	Phest		
	[H+] trial		
10	a	60	
	b		
	c		
	d		
	e		
15	g'([H+])	65	
20	[Na+]	70	
	Nb		
	C		
	Va		
	MaVa		
25	control	75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 026 TOT. REG. 84 USER MODE

ENG ____ FIX 3 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		ACID	BASE
02		Volume added = 0	Volume added ≠ 0

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

WEAK ACID/BASE TITRATION CURVE

PROGRAM REGISTERS NEEDED: 59

HEWLETT PACKARD
SOLUTION BOOK:
CHEMICAL ENGINEERING

ROW 1 (1 - 3)

ROW 2 (3 - 9)

ROW 3 (9 - 17)

ROW 4 (17 - 22)

ROW 5 (23 - 29)

ROW 6 (30 - 36)

ROW 7 (36 - 39)

ROW 8 (39 - 41)

ROW 9 (42 - 51)

ROW 10 (51 - 60)

ROW 11 (61 - 68)

ROW 12 (69 - 76)

ROW 13 (77 - 83)

ROW 14 (83 - 91)

ROW 15 (92 - 99)

ROW 16 (100 - 107)

ROW 17 (108 - 115)

ROW 18 (115 - 124)

ROW 19 (124 - 133)

ROW 20 (133 - 141)

ROW 21 (141 - 148)

ROW 22 (149 - 158)

ROW 23 (159 - 166)

ROW 24 (166 - 172)

ROW 25 (172 - 183)

ROW 26 (184 - 191)

ROW 27 (192 - 201)

ROW 28 (201 - 208)

ROW 29 (209 - 216)

ROW 30 (217 - 224)

ROW 31 (224 - 233)

ROW 32 (234 - 236)

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026

Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443

Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

CHEMICAL ENGINEERING

STRAIGHT FIN EFFICIENCY
CONSERVATION OF ENERGY
HYDROCARBON COMBUSTION
HEAT TRANSFER THROUGH COMPOSITE CYLINDERS AND WALLS
VON KÁRMÁN ANALOGY FOR HEAT AND MASS TRANSFER
EQUATIONS OF STATE
REVERSIBLE POLYTROPIC PROCESS FOR AN IDEAL GAS
CONDUIT FLOW
FLUID TRANSPORT NUMBERS
SINGLE STAGE EQUILIBRIUM FLASH CALCULATION
WEAK ACID/BASE TITRATION CURVE

HEWLETT PACKARD

00041-90100 Rev. D 5/83 Printed in U.S.A.